# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.
#
# Old Maintainer: Sergej Pupykin <arch+pub@sergej.pp.ru>

pkgname=bluez-hciconfig
_pkgname=bluez
pkgver=5.65
pkgrel=1
pkgdesc='Deprecated hciconfig tool from bluez.'
url='http://www.bluez.org/'
arch=('x86_64' 'aarch64')
license=('GPL2')
depends=('bluez' 'bluez-utils' 'bluez-libs' 'bluez-tools')
source=("https://www.kernel.org/pub/linux/bluetooth/bluez-$pkgver.tar.xz")
sha512sums=('c20c09a1a75053c77d73b3ce15ac7fd321eb6df5ca1646d57c6848b87c0c9957908bc17dd928da4ef2aacfc8667877cbc7511c1ba43db839bfa9bf1fb8269907')

build() {
  cd "$_pkgname-$pkgver/tools"

  gcc hciconfig.c parser/{csr,parser,lmp}.c -lbluetooth -o \
    hciconfig -DVERSION=\"$pkgver\" -I..

  #gcc hciconfig.c parser/{csr,parser,lmp}.c ../src/textfile.c -lbluetooth -o \
  #  hciconfig -DVERSION=\"$pkgver\" -I..
}

package() {
  cd "$_pkgname-$pkgver/tools"

  install -Dm 755 hciconfig "$pkgdir/usr/bin/hciconfig"
  install -Dm 755 hciconfig.1 "$pkgdir/usr/share/man/man1/hciconfig.1"
}

